Modellering med 1.ordens lineære differentialligninger

rev. 02.11.16. KS

```
> restart:with(plots):
```



1. Problem→model→løsning

På figuren ser vi en beholder som er fuld af vand tilsat salt. Til tiden t har opløsningen koncentrationen $x(t)$ overalt i beholderen. Der sker en indstrømning i beholderen af saltholdigt vand med konstant koncentration k. Indstrømninghastigheden (som fx måles i Liter/Sek) betegnes v. Der sker en tilsvarende udstrømning med strømningshastighed v.

Saltkoncentrationen i beholderen ved starttidspunktet kalder vi $x0$, der gælder altså $x(0) = x0$. Nu vil vi meget gerne vide hvordan saltkoncentrationen $x(t)$ udvikler sig! Men dette kan vi ikke trække ud af de givne informationer. Vi er nødt til at gå indirekte frem. Det viser sig muligt at tænke sig frem til et udtryk for den hastighed $x'(t)$ hvormed saltkoncentrationen i beholderen ændres, og dette udtryk kan vi arbejde videre med bagefter.

Opgave 1

Lad Δt betegne et lille tidsinterval og besvar følgende spørgsmål:
1) Hvor meget vand strømmer der ind/ud i løbet af Δt?
2) Hvor meget salt strømmer der ind i beholderen i løbet af Δt?
3) Hvor meget salt strømmer der ud af beholderen i løbet af Δt (vi antager at udløbsvandets

koncentration er $x(t)$)?
4) Hvor stor er ændringen Δx i beholderens saltkoncentration i løbet af Δt (beholderens volumen sættes til 1)?

Løsning

1) $v \cdot \Delta t$
2) $v \cdot \Delta t \cdot k$
3) $v \cdot \Delta t \cdot x(t)$
4) $\Delta x = \dfrac{v(k - x(t)) \cdot \Delta t}{1} = v(k - x(t)) \cdot \Delta t$

Opgave 2

Find på basis af opgave 1 spørgsmål 4) et udtryk for $x'(t)$, og indsæt det i stedet for XX i nedenstående kommando:

```
> lign1:=diff(x(t),t)=XX;
```

$$lign1 := \frac{d}{dt}\, x(t) = XX \tag{1.1}$$

Løsning

Fra opgave 1.4 får vi et udtryk for den gennemsnitshastighed hvormed koncentrationen har ændret sig:
$\dfrac{\Delta x}{\Delta t} = v(k - x(t))$.

Da $x'(t)$ = grænseværdien af $\dfrac{\Delta x}{\Delta t}$, har vi hermed det ønskede udtryk for $x'(t)$:

```
> lign1:=diff(x(t),t)=v*(k-x(t));
```

$$lign1 := \frac{d}{dt}\, x(t) = v\,(k - x(t)) \tag{1.2.1}$$

Det er nu lykkedes os at opstille en *differentialligning*, hvor saltkoncentrationen $x(t)$ optræder som ukendt funktion.
Vi kan finde et foreløbigt udtryk for $x(t)$ ved at at indsætte ligningen i "dsolve":

```
> dsolve(lign1);
```

$$x(t) = k + e^{-vt}\, \_C1 \tag{1.2}$$

Opgave 3

Antag nu at

```
> k:=1/20:
  v:=1/10:
  x0:=1/10:
```

Herefter har differentialligningen formen:

```
> dsolve(lign1);
```

$$x(t) = \frac{1}{20} + \mathrm{e}^{-\frac{t}{10}} \_C1 \quad \textbf{(1.3)}$$

Find et endeligt udtryk for $x(t)$, og plot $x(t)$ i intervallet $t \in [0\,;50]$. Kommentér plottet: Hvordan udvikler saltvandskoncentrationen sig i beholderen?
Vink: Vi kan først plotte, når vi også kender konstanten $\_C1$. Den kan findes vha. *begyndelsesbetingelsen* $x(0) = \mathrm{x0}$.

Løsning

```
> dsolve(lign1):
  subs(t=0,rhs(%))=x0:
  _C1:=solve(%,_C1);
```

$$\_C1 := \frac{1}{20} \quad \textbf{(1.3.1)}$$

Hermed er konstanten bestemt. Vi kunne også at fundet $x(t)$ i ét hug således:

```
> dsolve({lign1,x(0)=x0});
  sol1:=rhs(%):
```

$$x(t) = \frac{1}{20} + \frac{\mathrm{e}^{-\frac{t}{10}}}{20} \quad \textbf{(1.3.2)}$$

Vi plotter højresiden sammen med den vandrette linje gennem $\left(0, \frac{1}{20}\right)$:

```
> plot([sol1,1/20],t=0..50,color=[red,blue],view=0..0.10);
  v:='v':x0:='x0':k:='k':
```



Kommentar:

Saltkoncentrationen i beholderen er fra start $x0=\frac{1}{10}$ men nærmer sig efterhånden det indstrømmende vands saltkoncentration $k=\frac{1}{20}=0.05$. Faldhastigheden er størst i begyndelsen og aftager gradvist.

2. Videregående undersøgelser via modellen

Ved en **1. ordens lineær differentialligning** forstås en ligning af typen:

```
> dlign:=diff(x(t),t)+p(t)*x(t)=q(t);
```

$$dlign := \frac{\mathrm{d}}{\mathrm{d}t}\, x(t) + p(t)\, x(t) = q(t) \tag{2.1}$$

hvor $p(t)$ og $q(t)$ er givne kontinuerte funktioner.
Ligningen kaldes *homogen* hvis $q(t)=0$, og ellers *inhomogen.*

Opgave 4

I afsnit 1 fik vi opstillet ligningen:

```
> lign1;
```

$$\frac{d}{dt} x(t) = v (k - x(t)) \tag{2.2}$$

Vis at "lign1" er af typen "dlign", og bestem hvad $p(t)$ og $q(t)$ svarer til. Er "lign1" homogen eller inhomogen?

Løsning

```
> lign1:=diff(x(t),t)=v*(k-x(t));
```

$$lign1 := \frac{d}{dt} x(t) = v (k - x(t)) \tag{2.1.1}$$

kan omskrives til

```
> lign1:=diff(x(t),t)+v*x(t)=v*k;
```

$$lign1 := \frac{d}{dt} x(t) + v\, x(t) = v\, k \tag{2.1.2}$$

Hvis vi sætter $p(t) = v$ og $q(t) = v\, k$ ses det at "lign1" er af formen "dlign".
NB: Vi har arbejdet med en førsteordens lineær differentialligningen med *konstante koefficienter* !
Hvis enten $v = 0$ eller $k = 0$, er "lign1" homogen. I modsat fald inhomogen.

Vi betragter igen forsøgsopstillingen fra afsnit 1. Vi ser på det tilfælde at det vand der strømmer ind i beholderen, er rent (uden salt, dvs. $k = 0$), men at ind- og udstrømningshastigheden til gengæld er givet ved en variabel kontinuert funktion $v(t)$.

Opgave 5

Vis at der til den nye forsøgsopstilling svarer en vaskeægte **homogen** førsteordens lineær differentialligning. Opskriv den ved at udfylde nedenstående kommando. Løs derefter ligningen med "dsolve".

```
> lign2:=diff(x(t),t)+XX=0;
```

$$lign2 := \frac{d}{dt} x(t) + XX = 0 \tag{2.3}$$

Løsning

```
> lign2:=diff(x(t),t)+v(t)*x(t)=0;
```

$$lign2 := \frac{d}{dt} x(t) + v(t)\, x(t) = 0 \tag{2.2.1}$$

Den homogene ligning! Eneste at bemærke er at $p(t)$ i dette eksempel hedder $v(t)$.
Den *generelle løsningsformel* for den homogene ligning er:

```
> dsolve(subs(q(t)=0,dlign));
```

$$x(t) = \_C2\, e^{\int -p(t)\, dt} \tag{2.2.2}$$

Her skal vi altså i vores eksempel blot erstatte $p(t)$ med $v(t)$.

Opgave 6

Antag nu at

```
> v:=t->(cos(t)+1)/10;
  x0:=1/10:
```

$$v := t \mapsto \frac{\cos(t)}{10} + \frac{1}{10} \tag{2.4}$$

Find et endeligt udtryk for $x(t)$, og plot $x(t)$ i intervallet $t \in [0\,;50]$. Kommentér plottet.

Løsning

```
> lign2;
```

$$\frac{d}{dt}\,x(t) + \left(\frac{\cos(t)}{10} + \frac{1}{10}\right) x(t) = 0 \tag{2.3.1}$$

```
> dsolve({lign2,x(0)=x0});
  sol2:=rhs(%):
```

$$x(t) = \frac{e^{-\frac{t}{10} - \frac{\sin(t)}{10}}}{10} \tag{2.3.2}$$

```
> plot(sol2,t=0..50);
  v:='v':x0:='x0':
```



Koncentrationen falder fra begyndelsensværdien 1/10 mod 0. Faldhastigheden er størst i begyndelsen og aftager gradvist. Indstrømningshastigheden er $\frac{1}{10}\cos(t)+\frac{1}{10}$ og svinger derfor mellem 0 og $\frac{2}{10}$. Vi ser derfor en række vandrette hældningskoefficienter på kurven, hvor den indstrømmende koncentration er 0.

Vi vender endnu en gang tilbage til forsøgsopstillingen fra afsnit 1. Vi forudsætter igen at det indstrømmende vand har konstant saltkoncentration *k*. Endvidere at ind- og udstrømningshastigheden er en variabel kontinuert funktion $v(t)$.

Opgave 7

Vis at der til denne nye forsøgsopstilling svarer en almindelig **inhomogen** førsteordens lineær differentialligning. Opskriv den ved at udfylde nedenstående kommando. Løs derefter ligningen med "dsolve".

```
> lign3:=diff(x(t),t)+XX=0;
```

$$lign3 := \frac{\mathrm{d}}{\mathrm{d}t}\, x(t) + XX = 0 \qquad (2.5)$$

Løsning

```
> lign3:=subs(v=v(t),lign1);
```

$$lign3 := \frac{d}{dt} x(t) + v(t)\, x(t) = v(t)\, k \qquad \textbf{(2.4.1)}$$

Den inhomogene ligning! Eneste at bemærke er at $p(t)$ i vores eksempel hedder $v(t)$, og at $q(t) = k \cdot v(t)$.

Den *generelle løsningsformel* for den inhomogene ligning er:

```
> dsolve(dlign);
```

$$x(t) = \left(\int q(t)\, e^{\int p(t)\, dt}\, dt + \_C2 \right) e^{\int -p(t)\, dt} \qquad \textbf{(2.4.2)}$$

Her skal vi altså i vores eksempel blot erstatte $p(t)$ med $v(t)$ og $q(t)$ med $k \cdot v(t)$.

```
> dsolve(subs({p(t)=v(t),q(t)=k*v(t)},dlign));
```

$$x(t) = \left(k\, e^{\int v(t)\, dt} + \_C2 \right) e^{\int -v(t)\, dt} \qquad \textbf{(2.4.3)}$$

```
> dsolve(lign3);
```

$$x(t) = \left(k\, e^{\int v(t)\, dt} + \_C2 \right) e^{\int -v(t)\, dt} \qquad \textbf{(2.4.4)}$$

Opgave 8

Antag nu at

```
> k:=1/20:
  v:=t->(cos(t)+1)/10;
  x0:=1/10;
```

$$v := t \mapsto \frac{\cos(t)}{10} + \frac{1}{10}$$

$$x0 := \frac{1}{10} \qquad \textbf{(2.6)}$$

Find et endeligt udtryk for $x(t)$, og plot $x(t)$ i intervallet $t \in [0\,;50]$. Kommentér plottet.

Løsning

```
> lign3;
```

$$\frac{d}{dt} x(t) + \left(\frac{\cos(t)}{10} + \frac{1}{10} \right) x(t) = \frac{\cos(t)}{200} + \frac{1}{200} \qquad \textbf{(2.5.1)}$$

```
> dsolve({lign3,x(0)=x0});
  sol3:=rhs(%):
```

$$x(t) = \frac{1}{20} + \frac{e^{-\frac{t}{10} - \frac{\sin(t)}{10}}}{20} \qquad \textbf{(2.5.2)}$$

```
> plot([sol3,0.05],t=0..50,color=[red,blue],view=0..0.10);
```



Plottet ligner i høj grad det forrige plot. Vigtigste forskel er at beholderens saltkoncentration ikke tenderer mod 0, men mod det indstrømmende vands saltkoncentration $k = 0.05$. Den svingende indstrømningshastighed medfører de samme vandrette tangenter som før.

Opgave 9

At der er én særlig løsning for en given begyndelsesværdi $x0$, kan vi tydeliggøre ved at plotte løsninger for *lign3* svarende til forskellige begyndelses-saltkoncentrationen i beholderen. Gør det hvor nogle er mindre og andre større end det indstrømmende vands saltkoncentration 0.05.

Løsning

```
> x:='x':
  x0:='x0':
  dsolve({lign3,x(0)=x0}):
  plot([seq(rhs(%),x0=0..0.15,0.01)],t=0..50);
```


0.14
0.12
0.10
0.08
0.06
0.04
0.02
0
10
20
30
40
50
t